

NAG Library Routine Document

G13ACF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G13ACF calculates partial autocorrelation coefficients given a set of autocorrelation coefficients. It also calculates the predictor error variance ratios for increasing order of finite lag autoregressive predictor, and the autoregressive parameters associated with the predictor of maximum order.

2 Specification

```
SUBROUTINE G13ACF (R, NK, NL, P, V, AR, NVL, IFAIL)
  INTEGER          NK, NL, NVL, IFAIL
  REAL (KIND=nag_wp) R(NK), P(NL), V(NL), AR(NL)
```

3 Description

The data consist of values of autocorrelation coefficients r_1, r_2, \dots, r_K , relating to lags $1, 2, \dots, K$. These will generally (but not necessarily) be sample values such as may be obtained from a time series x_t using G13ABF.

The partial autocorrelation coefficient at lag l may be identified with the parameter $p_{l,l}$ in the autoregression

$$x_t = c_l + p_{l,1}x_{t-1} + p_{l,2}x_{t-2} + \dots + p_{l,l}x_{t-l} + e_{l,t}$$

where $e_{l,t}$ is the predictor error.

The first subscript l of $p_{l,l}$ and $e_{l,t}$ emphasizes the fact that the parameters will in general alter as further terms are introduced into the equation (i.e., as l is increased).

The parameters are determined from the autocorrelation coefficients by the Yule–Walker equations

$$r_i = p_{i,1}r_{i-1} + p_{i,2}r_{i-2} + \dots + p_{i,i}r_{i-i}, \quad i = 1, 2, \dots, l$$

taking $r_j = r_{|j|}$ when $j < 0$, and $r_0 = 1$.

The predictor error variance ratio $v_l = \text{var}(e_{l,t}) / \text{var}(x_t)$ is defined by

$$v_l = 1 - p_{l,1}r_1 - p_{l,2}r_2 - \dots - p_{l,l}r_l.$$

The above sets of equations are solved by a recursive method (the Durbin–Levinson algorithm). The recursive cycle applied for $l = 1, 2, \dots, (L-1)$, where L is the number of partial autocorrelation coefficients required, is initialized by setting $p_{1,1} = r_1$ and $v_1 = 1 - r_1^2$.

Then

$$\begin{aligned} p_{l+1,l+1} &= (r_{l+1} - p_{l,1}r_l - p_{l,2}r_{l-1} - \dots - p_{l,l}r_1) / v_l \\ p_{l+1,j} &= p_{l,j} - p_{l+1,l+1}p_{l,l+1-j}, \quad j = 1, 2, \dots, l \\ v_{l+1} &= v_l(1 - p_{l+1,l+1})(1 + p_{l+1,l+1}). \end{aligned}$$

If the condition $|p_{l,l}| \geq 1$ occurs, say when $l = l_0$, it indicates that the supplied autocorrelation coefficients do not form a positive definite sequence (see Hannan (1960)), and the recursion is not continued. The autoregressive parameters are overwritten at each recursive step, so that upon completion the only available values are $p_{L,j}$, for $j = 1, 2, \dots, L$, or $p_{l_0-1,j}$ if the recursion has been prematurely halted.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

Durbin J (1960) The fitting of time series models *Rev. Inst. Internat. Stat.* **28** 233

Hannan E J (1960) *Time Series Analysis* Methuen

5 Arguments

- 1: R(NK) – REAL (KIND=nag_wp) array *Input*
On entry: the autocorrelation coefficient relating to lag k , for $k = 1, 2, \dots, K$.
- 2: NK – INTEGER *Input*
On entry: K , the number of lags. The lags range from 1 to K and do not include zero.
Constraint: $NK > 0$.
- 3: NL – INTEGER *Input*
On entry: L , the number of partial autocorrelation coefficients required.
Constraint: $0 < NL \leq NK$.
- 4: P(NL) – REAL (KIND=nag_wp) array *Output*
On exit: $P(l)$ contains the partial autocorrelation coefficient at lag l , $p_{l,l}$, for $l = 1, 2, \dots, NVL$.
- 5: V(NL) – REAL (KIND=nag_wp) array *Output*
On exit: $V(l)$ contains the predictor error variance ratio v_l , for $l = 1, 2, \dots, NVL$.
- 6: AR(NL) – REAL (KIND=nag_wp) array *Output*
On exit: the autoregressive parameters of maximum order, i.e., p_{Lj} if $IFAIL = 0$, or $p_{l_0-1,j}$ if $IFAIL = 3$, for $j = 1, 2, \dots, NVL$.
- 7: NVL – INTEGER *Output*
On exit: the number of valid values in each of P, V and AR. Thus in the case of premature termination at iteration l_0 (see Section 3), NVL is returned as $l_0 - 1$.
- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $NK \leq 0$,
or $NL \leq 0$,
or $NK < NL$.

$IFAIL = 2$

On entry, the autocorrelation coefficient of lag 1 has an absolute value greater than or equal to 1.0; no recursions could be performed.

$IFAIL = 3$

Recursion has been prematurely terminated; the supplied autocorrelation coefficients do not form a positive definite sequence (see Section 3). Argument NVL returns the number of valid values computed.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in *How to Use the NAG Library and its Documentation* for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in *How to Use the NAG Library and its Documentation* for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in *How to Use the NAG Library and its Documentation* for further information.

7 Accuracy

The computations are believed to be stable.

8 Parallelism and Performance

G13ACF is not threaded in any implementation.

9 Further Comments

The time taken by G13ACF is proportional to $(NVL)^2$.

10 Example

This example uses an input series of 10 sample autocorrelation coefficients derived from the original series of sunspot numbers generated by the G13ABF example program. The results show five values of each of the three output arrays: partial autocorrelation coefficients, predictor error variance ratios and autoregressive parameters. All of these were valid.

10.1 Program Text

```

Program g13acfe

!      G13ACF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g13acf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                    :: i, ifail, nk, nl, nvl
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ar(:), p(:), r(:), v(:)
!      .. Executable Statements ..
Write (nout,*) 'G13ACF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) nk, nl

Allocate (ar(nl),p(nl),r(nk),v(nl))

!      Read in data
Read (nin,*)(r(i),i=1,nk)

!      Calculate partial ACF
ifail = -1
Call g13acf(r,nk,nl,p,v,ar,nvl,ifail)
If (ifail/=0) Then
  If (ifail==3) Then
    Write (nout,99999) ' Only', nvl, 'valid sets were generated'
    Write (nout,*)
  Else
    Go To 100
  End If
End If

!      Display results
Write (nout,*) 'Lag Partial Predictor error Autoregressive'
Write (nout,*) ' autocorrn variance ratio' // ' parameter'
Write (nout,*)
Write (nout,99998)(i,p(i),v(i),ar(i),i=1,nvl)

100 Continue

99999 Format (1X,A,I2,A)
99998 Format (1X,I2,F9.3,F16.3,F14.3)
End Program g13acfe

```

10.2 Program Data

G13ACF Example Program Data

```

10 5
  0.8004    0.4355    0.0328   -0.2835   -0.4505
-0.4242   -0.2419   -0.0550    0.3783    0.5857

```

10.3 Program Results

G13ACF Example Program Results

Lag	Partial autocorrn	Predictor error variance ratio	Autoregressive parameter
1	0.800	0.359	1.108
2	-0.571	0.242	-0.290
3	-0.239	0.228	-0.193
4	-0.049	0.228	-0.014
5	-0.032	0.228	-0.032

This plot shows the partial autocorrelations for all possible lag values. Reference lines are given at $\pm z_{0.975}/\sqrt{n}$.

